BOOKING OPERATIONS IN A JAIL: A GPSS MODEL

Val Silbey and
William M. Rand

Florida International University

ABSTRACT

This study used a GPSS model of the booking operations in a county jail to determine the effect of randomness and manpower allocations on the processing of incoming inmates. Problems of data collection and alternative presentation to management are considered. The model illustrated that the major delays and manpower requirements for booking resulted from factors not controllable by jail management.

INTRODUCTION

This paper presents the results of the modelling effort in a large metropolitan county jail. The focus on only one facet, that of bookings, resulted from the stated purpose of identifying the trade-offs and bottlenecks in the initial processing of incoming inmates.

One is struck by the realization that the primary focus of most simulation exercises lies on the modelling and output results activities. However, considering the processes involved, it appears that these two elements comprise only a small fraction of a simulation project. This paper describes the analysis of a booking operation in a county jail with some concern on the data collection and presentation of results to decision makers.

The model itself requires few advanced concepts (with the exception of the split block, the logic and syntax of the GPSS model can be taught in two or three classroom sessions) but, as is normally the case, data collection for the model presents some problems. In addition, in order to communicate the structure and behavior of the model and to "sell" the results to management, effort needs to be expended to put it into their perspective.

This report provides first an explanation of what the various steps of the booking process are. This explanation is followed by a description of the GPSS model. After the model has been discussed, the results of the various simulation runs and their implications are presented.

DESCRIPTION OF BOOKING PROCESS

The booking process (see Figure 1) in the County Jail begins when a vehicle arrives in the parking yard of the jail. The yard is totally enclosed by four walls and entry is granted by a corrections officer in a control booth. Access to the yard is permitted through an electrically operated gate that rolls up into the roof. Once the vehicle is inside the yard, the gate is lowered behind it.



FIGURE 1 - INMATE FLOW THROUGH BOOKING OPERATIONS

Having disembarked from the vehicle, the police officer(s) and his prisoner(s) proceed to the outer door of the sally port. The outer door of the sally port is cranked open by the corrections officer in the control booth. The police officer places the handcuffed prisoner in the sally port and the outer door is closed behind him. Once the prisoner is in the sally port and the door shut safely behind him, the police officer, still in the yard, surrenders all of his weapons to the control booth officer. All guns, mace, clubs, etc., are passed to the booth officer through a window. After all weapons have been surrendered, the outer door of the sally port is again cranked open, the police officer joins his prisoner in the sally port, and the outer door is shut behind them.

As this procedure is taking place, it is witnessed by the rear lobby officer. He is the only man in the rear lobby of the jail who has the key to the inner door of the sally port.

Once the police officer and the prisoner are in the sally port, the rear lobby jailer opens the inner door to grant admission to the rear lobby of the jail. Once inside, this "Key Man" closes and locks the door.

The police officer and the inmate are directed to the frisk search table. The police officer removes the handcuffs from the prisoner and proceeds to booking officer number one (BOOK1 in the simulation model) to begin the paperwork of the booking process.

As the police officer is starting his paperwork, a corrections officer begins the frisk search of the inmate. The inmate is instructed to remove everything from his pockets and to take off his shoes. He then places his hands on the edge of the frisk search table and the corrections officer backs his feet up until the inmate keeps his balance by maintaining his hands on the table. The inmate's feet are then spread approximately two feet apart, further adding to his instability. "The more precarious his position, the less likely he will be to throw a punch." The corrections officer then places his leg between the feet of the inmate in order to control him should the inmate decide to "throw a punch" in spite of his position. The officer begins by checking the collar, then the sleeves, the upper body, all pockets, the belt line, the crotch, both pants legs and cuffs, long hair, and the soles of the feet in succession. He also checks the insides of the shoes and their heels for weapons or other contraband. The prisoner is then placed in a rear hold cell (labelled Rear Hold Cell A in Figure 1).

One might think that this frisk search is redundant given that the inmate has been frisked by the police officers upon arrest. However, in the heat of the arrest, the frisk search is occasionally overlooked by the police and in any event, is generally not thorough enough.

Meanwhile, the police officer has begun the paperwork of the booking process by giving his arrest report to booking officer number one. The prisoner's name and other relevant information is determined from the arrest report and submitted by booking officer number one via a computer terminal to the National Crime Information Center. This entry is made to ascertain aliases, outstanding warrants, previous records and other relevant information. Upon completion of this initial processing, the police officer leaves the jail.

Almost all offenses, with few exceptions, are bondable. One reason for the check with the data base is to insure that there is not an outstanding warrant on an inmate that would require a higher bond than the offense for which he is currently being held. For example, if the inmate was arrested for aggravated assault, the bond would be $1,500. If the prisoner was also wanted on a murder warrant, without the data base check, he could bond out of jail for $1,500 when in fact, he is an alleged murderer which is an unbondable offense.

Booking officer number one logs the inmate, assigns a jail number to the inmate and passes the arrest report to booking officer number two (BOOK2 in the simulation model). Booking officer number two reads the arrest report and begins typing the jail card. The jail card is the inmate's official jail document and contains the inmate's name, address, date of birth, social security number, occupation, charge bond, (which booking officer number two determines with the aid of a computer printout of offenses and bonds) and mother's maiden name (mother's maiden name is used as an identification cross check on release to insure that another inmate does not prematurely exit the jail by claiming to be someone else).

When the jail card has been substantially completed, the second booking officer asks a rear lobby jailer to retrieve the inmate from the rear hold cell. The inmate is requested by booking officer number two to help complete the jail card by providing his social security number and mother's maiden name. The inmate is then assigned a booking time, advised of the charges and the applicable bond. He is led to an adjacent room for photographs and fingerprinting.

The identification officer (ID OFF in the simulation model) photographs and fingerprints the inmate. On completion of that task, the inmate is allowed the one proverbial phone call. That call has an official time limit, but operationally no limit is placed on the number or duration of the calls that the inmate makes. On completion of the calls, the inmate is placed in another rear hold cell.

At this point in the proceedings, a number of variations could occur. If the phone call, that the inmate made, was effective in raising the necessary bond for his release, the inmate is placed in one of two large dormitory cells also located on the ground floor, to await the arrival of his bondsman or other monied person.

If the inmate already has the appropriate amount of cash on his person, he is escorted by a rear lobby

officer to the front lobby of the jail where he is placed in a hold cell until the release desk officer can process his release.

If a bondsman is the vehicle for the inmate's release, on the bondsman's arrival in the outer front lobby of the jail, the inmate is retrieved and placed in a hold cell.

For an inmate who cannot raise the required bond, the procedure is different. His next step is the Pretrial Release and TASC interviews. The objective of the Pretrial Release Program is to help reduce the jail's population by attempting to identify arrested individuals who may not be able to raise bond for their release but otherwise are reasonable risks for release on their own recognizance.

The attempt here is to establish the relative seriousness of the crime and given that the crime qualifies for the program, elicit personal information from the suspect to establish his appearance in court should he be released. The interview touches on such areas as: does the man own his home, is he employed and for how long, is his family here in the County, is this his first arrest and other such questions that try to ascertain the man's stability as a County resident, and the probability that he might not appear for a scheduled court appearance.

The TASC Program (Treatment Alternatives to Street Crime) concerns itself with trying to identify persons who committed crimes because of their drug habits. The thrust is directed at young addicts rather than "smokers" or occasional users.

On completion of the two interviews, the inmate is escorted to the property room. He is advised to surrender all cash and any valuables such as jewelry, etc. This process is an effort to mitigate the temptation of other inmates to assault and rob a "wealthier" inmate.

After the property room processing, the inmate is placed in the final holding cells and the booking process is complete. He waits there until the next bond hearing convenes. The purpose of the bond hearing is to establish appropriate bond reductions or court releases on recognizances. The bond hearings do not impact on the manpower or administrative needs of the booking process and are mentioned only in the interest of identifying the end result of the booking process.

GPSS MODEL OF THE BOOKING PROCESS

The preceding description of the booking process has identified the various steps that have to be performed to book an inmate into the Jail. These steps have been translated into the GPSS language and define the logic of the model's operations. A listing of the GPSS Simulation model is given in Figure 2.

Key elements of the model are the correctional officers who process the inmate. These elements are listed below:

KEY

BOOK1 = Booking Officer Number One
BOOK2 = Booking Officer Number Two
REARO = Rear Lobby Officers
IDOFF = Identification Officer
PTRO = Pretrial Release and TASC Interviewers
PROPO = Property Room Officer

The line number on the left-hand side of the computer printouts are used to identify the steps of the model in the following discussion:

```
1       GENERATE        10,FN1
```

In this step, police officers and inmates are created at an interarrival rate of 10 minutes in accordance with Function 1 - the Poisson distribution. This means that the average time (in the long run) between the arrivals in the yard of inmates will be 10 minutes, i.e., an average of 6 inmates per hour would arrive to be processed. The randomness of the interarrival rate is determined by the Poisson distribution and interarrival rates are determined by the selection of random numbers.

```
6       ENTER       REARO
```

In this statement the "key man", if available, permits the inmate/police officer transaction to leave the sally port and enter the rear lobby. If the "key man" is not available, then both incoming inmate and police officers have to wait in the sally port. Assigning a priority of 5 to the transaction in the previous step (Step 5) states that the rear lobby officers (REARO) have to process the incoming transactions in the sally port before moving inmates out of the rear hold cells where their priority is lower (see Steps 26 and 53 where the priority of the incoming inmates is changed).

```
10      SPLIT       1,POLOF
```

In this statement the police officer is split from the incoming inmate and sent to BOOK1, where he forms a QUEUE (Step 15), a waiting line for the first booking officer.

In steps 11, 12, and 13 the inmate is frisk searched, a process that takes from 2 to 6 minutes, uniformly distributed. After the frisk search the inmate is placed in Rear Hold Cell A by transferring him to a waiting line called RHOLA (Transfer in Statement 14, waiting line in Statement 22).

In statement 18 (ADVANCE 6, 2) the charges are entered into the system. This entry of the arrest report takes from 4 to 8 minutes uniformly distributed over that range. After the completion of this process the police officer releases the correctional officer performing the entry (statement 19) and departs the system (statement 21).

After the initial paperwork is finished, (the match in statement 23), and if the second booking officer is available (statement 27), the rear lobby jailers bring the inmate to the desk for the preparation of the jail card. The fill-in of the jail card takes an average of five minutes with a range of ± 2 minutes (statement 25). (On some occasions the completion of the jail card may take longer; such as, on

New Year's Eve when a large number of the incoming inmates were intoxicated; however, such occurrences, while humorous, are rare and can be ignored when attempting to analyze the normal processing of incoming inmates.) Then the inmate moves to the fingerprinting and photography area and waits for the IDOFF (statement 37). For this movement to the ID area the rear lobby officer is not involved. Normally, the booking officer merely tells the incoming inmate where to go for fingerprints and photographs. Occasionally the inmate is assisted by a rear lobby jailer in this movement. But at all times when the process was observed, a rear lobby officer was standing by and observing the movement of the inmate. Although, it was not observed, rear lobby personnel stated that if the ID officer was becoming backlogged, inmates would be placed in the rear holding cells temporarily. Therefore the long queues that developed while waiting for the ID officer in some runs of the model (see discussion) would not be observed in actual operations. Based on actual observations, the work of the ID officer takes between 4 and 10 minutes with an average of 7 minutes (statement 40).

Following photographing and fingerprinting, the inmate makes his phone calls, which follow a poisson distribution with an average of 8 minutes (statement 45). The number of phone calls is unspecified in the model, since only the time duration is important.

The inmate then proceeds to the Pretrial Release and TASC interviews. In the model all inmates are interviewed, although in the actual processing of incoming inmates a large number of inmates are not eligible for either program. However, since the interviewers are not under the control of the Jail personnel, the model in effect tests the staffing requirement for this function given that all inmates are interviewed. The interviews themselves take from 15 to 25 minutes and are based primarily on structured questionnaires, designed to elicit the information previously described.

At this stage, the only remaining processing step is the check-in of inmate property with the property room, and the placement of the inmate into the two large dormitory cells. The model finishes with the placement of the inmate into these two cells (statement 65).

The model was run with different parameter values to test the utilization of the personnel assigned to rear lobby operations, to determine the effect of randomness, and to identify the factors affecting the total transit time of incoming inmates through the model. By this modelling effort, the analyst need not follow a group of inmates from start to finish, nor does it require experimentation with actual personnel assignments to determine their impact on the processing of inmates.

DATA COLLECTION ASPECTS

For data collection, the model provided the necessary conceptualization of what data should be collected. From this perspective, the importance of the modelling effort is not derived from the actual running of the model, but in defining the problem parameters in the first place. This function of modelling is discussed in some of the literature (1, 4, 7), but hardly touched upon in others (2, 5, 6). Yet this data definition function of modelling deserves emphasis since it provides a focus on controllable and uncontrollable elements of a system.

Naturally, what is needed to identify these parameters is not part of the regular data collection of jail operations. True, a log of inmate jail number assignments is kept, however, with booking also occurring in outlying police stations which receive a jail number from the main jail, using the log of jail numbers leads to an overloading of the system. The timing of inmate inflows, i.e., when did inmates arrive, is not part of the recording function. The actual timing of each of the operations had to be determined from observation. Again, a stopwatch study is typically a reactive measurement (if permitted). By talking to the correctional officers who handled the processing and by keeping track of actual performance, a rough estimate for each time consumption can be determined. In this case, since time is expressed in minutes, and since the number of data points is relatively small, the uniform distribution appeared to be a good approximation of actual behavior. But, even this data collection takes time and required more effort than structuring the model.

Secondly, for the preparation of reports, the data presented by the actual GPSS runs is too voluminous and confusing to somebody not familiar with GPSS. However, in order to determine which runs to make, it is necessary to identify the real objectives instead of the stated objectives. On the basis of discussion with jail management the budget situation appeared to be a dominant element, which in terms of model parameters can be transformed into personnel availability. Hence, alternatives with respect to number of correctional officers required to man booking positions needed to be run. In addition, court ordered mandates for the speedy processing of inmates provided another focus for the presentation of results. Finally, the recurrent comments of the processing personnel to the peak and slack periods required an analysis of arrival characteristics. Hence, for report preparation, additional determination of the perspective of management and operations personnel is required in order to provide the proper focus. Therefore, the specific trade-offs required are an identification of the number of correctional officers required to process the arrivals and the impact ("sensitivity") of the number of officers on the total transit time of the arrivals. Since effective utilization of personnel is a concern, the impact of varying rates of incoming inmates on the bottleneck processing steps is also of concern. This function of data collection, i.e., determination of what runs are

needed to satisfy decision makers, requires further emphasis in the simulation literature since it lays a foundation for the presentation of results.

RESULTS AND DISCUSSION

The following presentation discusses these sensitivity aspects from the necessary management perspective (3, 8). It should be noted, that the actual runs, with their plethora of detail, while available for backup (and as "boilerplate" to impress management) are much too voluminous to be presented for discussion. Therefore, one of two alternatives, either tabular or graphic presentation (depending on the stylistic orientation of management) has to be considered. Since that dilemma is unresolved, the following discussion is based on both the tabular and graphic material presented below.

A) EFFECT OF RANDOMNESS AND ARRIVAL RATE ON INMATE TRANSIT TIME

The first set of runs were designed to test the impact of variations in the number of arrivals into the system. This variation was generated by changing the average interarrival rate for inmates and by changing the random number generator for the different arrival rates. The results of the initial runs have been gathered in Table 1 and show the impact of the average number of arrivals per hour compared to the average transit time of all inmates completing the process. The same information has been plotted in figure 3. As can be seen from the figure, an apparent break occurs when the interarrival rate increases to 6 or more arrivals per hour. At 4 or 5 inmates arriving per hour, the system shows little variation in the transit time of the inmates, but with six or more arrivals the impact of randomness increases considerably.

	AVERAGE ARRIVAL VOLUME PER HOUR				
	4	5	6	7.5	10
	59.82	71.93	79.35	132.45	204.77
TRANSIT	60.70	73.90	92.06	138.74	269.14
TIMES	63.80	78.14	102.36	143.58	269.37
IN MINUTES	64.77	78.23	103.25	174.35	287.81
WITH	67.01	82.97	110.13	222.10	292.29
DIFFERENT	68.24	83.55	120.78	225.83	295.65
RANDOM	69.68	85.52	138.41	230.68	297.58
NUMBERS	79.05	89.66	139.20	249.56	297.62
AVERAGE	66.63	80.49	110.69	189.66	276.78
STANDARD DEVIATION	6.09	6.00	21.17	47.59	31.36
RANGE	19.23	16.73	59.85	117.11	92.85

TABLE 1 - IMPACT OF ARRIVAL VOLUME ON TRANSIT TIME



FIGURE 3 - EFFECT OF VOLUME ON TRANSIT TIME

The two lines in figure 3 show the linear regression of transit times on the number of arrivals. With the difference in system reaction patterns, the relative level of number of arrivals splits the inmate transit times into two categories. Hence, two regression lines, one for an average of 4, 5, and 6 arrivals per hour and another for 6, 7.5 and 10 arrivals per hour are needed. The slopes of the two lines give the sensitivity of transit time to volume. Under light volume assumptions, the increase of one arrival per hour (from 4 to 5 or from 5 to 6) increases the average inmate transit time (the time it takes an inmate from his initial arrival until his placement in cells 1A and 1B) by 22 minutes. Under heavy loading conditions, the average inmate transit time increases by 38.5 minutes for each increase of 1 inmate in the arrival rate. However, with the small amount of variation in the inmate process times for the light periods, the confidence in the 22 minute trade-off is much greater than in the 38.5 minute increase. For light loading of the system, the transit time ranged from about 1 hour to approximately 2 hours. However, at 10 arrivals per hour, the transit times ranged from just over

three hours to almost 5 hours. In other words, the difference for heavy loading periods of two hours indicates a higher impact due to randomness than the one hour difference in the light loading period.

B) EFFECT OF RANDOMNESS AND ARRIVAL RATE ON REAR LOBBY OFFICER UTILIZATION.

In the preceding section, the effect of changes in arrival rate on inmate transit times was discussed. The present allocation of personnel to the rear lobby processing was used as a basis for the next series of runs which tested the utilization of personnel:

BOOK1 -- 1 officer
BOOK2 -- 1 officer
IDOFF -- 1 officer
REARO -- 3 officers
PROPO -- 1 officer

The average utilization of these seven people, and the respective standard deviations is given in Table 2 (where M is mean and SD stands for Standard Deviation).

TABLE 2 - AVERAGE UTILIZATION OF CORRECTIONAL OFFICERS IN INCOMING INMATE PROCESSING

AVERAGE ARRIVAL RATE PER HOUR

FUNCTION	4 M/SD	5 M/SD	6 M/SD	7.5 M/SD	10 M/SD
BOOK1	.432/.046	.549/.039	.632/.039	.811/.093	.984/.031
BOOK2	.366/.041	.456/.032	.516/.031	.669/.062	.806/.033
IDOFF	.500/.050	.628/.039	.725/.042	.884/.072	.967/.012
REARO (3)	.312/.031	.403/.025	.445/.022	.508/.036	.615/.032
PROPO	.337/.034	.434/.034	.456/.021	.465/.017	.478/.017

These data indicate that relatively speaking the ID officer is the busiest individual in the rear lobby. However, as the number of incoming inmates increases, the function of BOOK1 becomes more important in setting the pace for the remaining processing. As BOOK1 becomes overloaded (any average utilization rate greater than .8--(80% is an indicator that not enough slack exists for effective scheduling and immediate adjustment to unusual requirements), the number of incoming inmates is queued up waiting for the initial preparation of their arrest forms. Therefore, the percentage of inmates completing the processing during their arrival shift declines. With the blockage in BOOK1, the succeeding processing steps are therefore also delayed.

If we compare shifts, the afternoon shift appears to show the heaviest loading with approximately 7.5 new arrivals per hour; the midnight shift has about 6 arrivals per hour and during the day time only approximately 4.5 new inmates arrive per hour. The two endpoints of the scale (10 arrivals per hour and 4 per hour) are included in the simulation to provide a better estimate of the relative impact of arrival rate.

The runs also indicated an internal cascading effect. If the processing of arrest forms is speeded up by adding another corrections officer, then during heavy loading periods, BOOK2 becomes a bottleneck and the waiting lines for IDOFF increase spectacularly. However, in all cases, the property room officer showed relatively modest utilization rates (less than .500) due to the intermediary placement of the pretrial release and TASC interviews. These interviews, for which personnel are not assigned by jail management, furnished the major bottleneck and time consumption in the processing of inmates.

The four phones that are available for inmate phonecalls are more than adequate. Utilization rates for these phones rarely exceed 25% (i.e., one phone used continuously) even under the heavy loading assumption of 10 arrivals per hour.

C) CHANGES IN NUMBER OF PERSONNEL ASSIGNED TO INMATE BOOKING FUNCTION.

Alternative assignments to the various functions were summarized as different employee allocation plans and presented for discussion. It should be noted that the assignment of manpower for the pretrial release and TASC (treatment alternative for street crime (PTRO in the model)), is not the responsibility of jail management, but is handled out of the state attorney's office. Therefore, the major changes in the model deal with the controllable elements, i.e., correctional officers in the booking process. Table 3 - Impact of manpower on transit time, and Figure 4 - Effect of manpower on transit time, illustrate the various alternatives. In addition, Table 3 also presents the utilization rate of the people performing the various functions. From these presentations, it can be seen that pretrial release and TASC, the one element outside the control of jail management with its heavy utilization, is the primary bottleneck in the speedy processing of inmates.

Alternative	NR of	REARO	PTRO*	BOOK1	BOOK2	IDOFF	Avg. Transit Time	% Arrivals Finished Processing	Incremental Annual Savings
1	5	2 .924	2 .970	1 .997	1 .776	1 .757	4:42	44.4	+ $11,000
2	6	2 .951	2 .970	1 .997	2 .802	2 .402	4:34	50.9	0
3	6	2 .961	2 .965	2 .562	1 .917	1 .919	5:41	48.9	0
4	5	2 .976	3 .924	1 .997	1 .778	1 .765	5:31	53.1	+ $11,000
5	6	2 .983	3 .929	1 .997	2 .402	1 .802	5:41	51.1	0
6	7	4 .485	2 .972	1 .995	1 .802	1 .812	5:24	47.2	- $11,000
7	6	3 .860	3 .934	1 .991	1 .814	1 .818	4:54	65	0
8	7	4 .591	3 .955	1 .991	1 .831	1 .795	3:24	71.7	- $11,000
9	7	3 .869	3 .965	1 .997	2 .419	1 .856	3:11	71.1	- $11,000
10	8	2 .973	2 .966	2 .578	2 .496	2 .468	5:43	44.4	- $22,000
11	8	3 .833	3 .967	1 .997	2 .415	2 .422	3:44	74.6	- $22,000
12	8	4 .591	3 .964	1 .997	2 .412	1 .815	3:28	70.1	- $22,000
13	9	3 .739	3 .905	2 .497	2 .426	2 .434	2:43	76.0	- $33,000
14	12	3 .895	3 .968	3 .405	3 .331	3 .331	3:37	70	- $66,000

* Note: PTRO assignments are not correctional officers.

TABLE 3: IMPACT OF MANPOWER ON TRANSIT TIME



FIGURE 4 - EFFECT OF MANPOWER ON TRANSIT TIME

The table identifies the number of correctional officers on the shift, the respective positions they occupied, how many occupied each position, the average utilization of each position and total transit time through the system. In addition, the percentage of new arrivals processed during a shift is given.

The last figure is either the positive or negative incremental savings given the level of manpower for a given run. Since booking positions are filled by experienced CO1's, incremental savings are calculated using the median annual salary for that rank, $11,000. As can be seen from the table, incremental savings vary around six correctional officers, six being the current level.

The runs range from a low of five employees to a high of twelve. Transit time ranges from four hours and fifty-one minutes at seven employees to three hours and twenty-eight minutes for eleven.

The logic for selection of possible position staffings is as follows: starting with the "six" norm, there was interest in what would happen if REARO was reduced from three to two. As can be seen, average utilization rose from .8598 to .9760. An interesting dichotomy exists here and is a lesson learned from production management. The more "efficiently" a resource is used (here, as measured by average utilization) the longer its queue becomes. So the trade-off becomes higher utilization versus increased queue length and resultant higher transit times through the system. This program however, results in a positive savings of one officer's salary per year or $11,000.

The preceding change from the norm brought on the subsequent area of interest; what will happen if in addition to reducing REARO to two officers, one also reduces PTRO to two? The cost in performance of that savings is a higher queue length and a consistently high transit time. Under alternative 1 utilization is already above 90 percent for the two categories that could conceivably still be reduced, hence the decision to add employees to the norm represented in alternative 3.

Alternative 4 adds one officer to BOOK2. This immediately cuts BOOK2 utilization in half. It also reduces the burden on REARO. In addition, transit time and queue lengths are reduced.

Alternative 5 utilizes six officers, but one officer has been shifted from BOOK2 to REARO. Utilization for BOOK2 is back in the eightieth percentile and REARO is now down from 80 percent to 59 percent. Transit time is lowered still more to three hours and twenty-four minutes.

In alternative 6 manpower requirements are basically the same as alternative 5 except that another officer is added to BOOK2. Once again, BOOK2 utilization drops almost in half, however, IDOFF utilization has risen slightly due to the increased output of BOOK2. Transit time and queue length remain relatively unchanged and negative savings is $11,000. This option probably would result in the least number of harried employees.

The above plans are immediate short run considerations for alterations to the current booking process. One will note however, that BOOK1 and IDOFF were not changed, principally because additional equipment expense would have to be incurred when increasing these positions; namely computer terminal and photographing equipment. Another BOOK1 officer and terminal will reduce BOOK1 utilization and help reduce the queues at that position. The trade-off here is queue length for fixed asset investments. It should be noted that most of the transaction time at BOOK1 is currently devoted to terminal response and print time. How much addition of a second terminal would reduce utilization at BOOK1 is of concern here.

In order to expand the IDOFF position, fixed asset investment would also be necessary in order to provide space for additional equipment. His utilization would probably drop significantly in return and lower his queue.

LARGER SYSTEM CONSIDERATIONS

All of the foregoing alternatives look at the shifting of manpower. The functions themselves have not been altered due primarily to the fact that most of the booking position activities are mandated by the courts or other legal considerations.

If manpower is reduced in the booking process, transit time will go up. If an inmate becomes irritable due to his wait in various queues, his resultant aggressive behavior could have a direct impact on the demeanor of the rest of the inmate population once he is placed with them. This will place added burdens on Corrections Department

decision makers to make decisions with respect to maintaining current levels of emotion within the inmate population. At a still higher level, it is not inconceivable that an inmate might challenge abnormally high transit times in the courts. This would cause the County Commission and their Manager to throw money at the problem and perhaps view the Corrections Department as a "problem area." This unfavorable tinge could create problems during future budget allocation hearings.

On the other hand, if transit time reduction is opted for by increasing manpower in the booking process, one possible way of neutralizing negative savings would be to draft a corrections officer from another part of the jail. Assuming acceptable utilization levels in the other systems, this will place additional strains on them. If, however, the cost of negative savings are born by the Correction Department Budget, the decision maker will have the task of searching for that money. If that search takes the form of additional requests, the County Manager at the behest of the Commission and the public, may have other priorities for that money.

SUMMARY

This paper has used the application of a GPSS simulation model to the booking process of a jail, to highlight two problem aspects not receiving their due in the literature. While the model itself is rather primitive, it does illustrate the usefulness of conceptualization prior to data collection and the use of results to support the decision makers in their understanding of the system.

The authors would be most anxious to receive reader input in three specific areas and one general area:

1. The data definition function of modelling
2. Determination of what runs are needed to satisfy decision makers.
3. The desirability of tabular or graphic presentation depending on the stylistic orientation of management
4. Any reader experiences with jail operations.

Naturally, any and all comments would be most appreciated.

REFERENCES

1. Emshoff, J.R., and Sisson, R.L., Design and Use of Computer Simulation Models, London, England: The Macmillan Company, 1970.

2. Fishman, G.S., Concepts and Methods in Discrete Event Digital Simulation, New York: John Wiley & Sons, 1973.

3. Little, J.D.C., "Models and Managers: The Concept of a Decision Calculus," Management Science, Vol. 16, No. 8, April 1970, pp. 466-485.

4. McKenney, J.L., "The Role of Computer Simulations in Planning," in Corporate Simulation Models, A.N. Schrieber, ed., Seattle, Wash.: Graduate School of Business Administration, University of Washington, 1970.

5. Naylor, T.H., Balintfy, J.L., Burdick, D.S., and Chu, K., Computer Simulation Techniques, New York: John Wiley & Sons, Inc., 1966.

6. Naylor, T.H., Computer Simulation Experiments with Models of Economic Systems, New York: John Wiley & Sons, 1971.

7. Teorey, T.J., "Validation Criteria for Computer Systems Simulation," Simulater, Vol. 6, No. 4, July 1975, pp. 9-20.

8. Woolsey, R.E.D., and Swanson, H.S., Operations Research for Immediate Application: A Quick & Dirty Manual, New York: Harper & Row Publishers, 1975 (Particularly Chapters 3 and 12).

```
      ****************************************************************
      *
      *   SIMULATION OF BOOKING OPERATIONS IN A JAIL
      *
      ****************************************************************
           1 FUNCTION   RN6,C24            POISSON FUNCTION
      0,0/0,1/0,104/0,2/0,222/0,3/0,355/0,4/0,509/0,5/0,69
      0,6/0,915/0,7/1,2/0,75/1,38/0,8/1,6/0,84/1,83/0,88/2,12
      0,9/2,3/0,92/2,52/0,94/2,81/0,95/2,99/0,96/3,2/0,97/3,5
      0,98/3,9/0,99/4,6/0,995/5,3/0,998/6,2/0,999/7/0,9997/8
      ****************************************************************
      *
      *   DESCRIPTION OF CAPACITIES
      *
      ****************************************************************
       SPORT STORAGE    30              SALLY PORT CAPACITY
       PHONE STORAGE    4               NUMBER OF PHONES
       REARO STORAGE    3               NUMBER OF REAR-LOBBY OFFICERS
       PTRO  STORAGE    2               NUMBER OF PTR INTERVIEWERS
      ****************************************************************
      *
      *   SET UP TABLES FOR DATA COLLECTION
      *
      ****************************************************************
           1 TABLE      M1,50,5,50      TABULATION OF TRANSIT TIME
           2 QTABLE     RHOLA,2,2,50    TABULATION OF TIME IN REARHOLDING CELL
           3 QTABLE     RHOLB,2,2,50    TABULATION OF TIME IN REARHOLDING CELL
           4 QTABLE     RHOLC,2,2,50    TABULATION OF TIME IN REARHOLDING CELL
      ****************************************************************
      *
      **    INCOMING PRISONERS
      *
      ****************************************************************
 1          GENERATE   10,FN1
 2          QUEUE      YARD
 3          ENTER      SPORT           ENTER SALLYPORT
 4          DEPART     YARD            ENTRY INTO BUILDING
 5          PRIORITY   5               ASSIGN PRIORITY FOR PROCESSING
 6          ENTER      REARO           GET A REAR LOBBY OFFICER
 7          LEAVE      SPORT           LEAVE SALLYPORT
 8          ADVANCE    1               TIME TO REMOVE INMATE FROM SALLYPORT
 9          LEAVE      REARO           INMATE NOW IN REAR LOBBY
10          SPLIT      1,POLOF         SEND POLICE OFFICER TO BOOKING
11          ENTER      REARO           GET A REARLOBBY OFFICER FOR INMATE
12          ADVANCE    4,2             FRISK SEARCH
13          LEAVE      REARO           REAR LOBBY OFFICER READY FOR NEXT INMATE
14          TRANSFER   ,WAITA          SEND INMATE TO REAR HOLDING CELL
      ****************************************************************
      *
      *   PREPARATION OF INMATE CHARGE REPORT AND JAILCARD
      *
      ****************************************************************
15   POLOF QUEUE      BOOKG           OFFICER WAITING FOR BOOKING
16          SEIZE      BOOK1           GET A BOOKING OFFICER
```

```
17          DEPART      BOOKG           DEPART QUEUE
18          ADVANCE     6,2             ENTER CHARGES INTO SYSTEM
19          RELEASE     BOOK1           CHARGE ENTRY FINISHED
20    PPRWK MATCH       INMTE           COORDINATE PAPERWORK
21          TERMINATE                   POLICE OFFICER FINISHED
22    WAITA QUEUE       RHOLA           REAR HOLDING CELLS
23    INMTE MATCH       PPRWK           MATCH PAPERWORK WITH INMATE
24          DEPART      RHOLA
25          QUEUE       RHOLB           INMATE WAITING FOR BOOKING
26          PRIORITY    3               LOWER PRIORITY FOR NEXT STEP
27          GATE NU     BOOK2           SEE IF BOOKING OFFICER IS AVAILABLE
28          ENTER       REARO           GET AN OFFICER FOR MOVEMENT
29          DEPART      RHOLB           LEAVE REAR HOLDING CELL
30          ADVANCE     1               PRISONER MOVEMENT TIME
31          LEAVE       REARO
      ***************************************************************
      *
      *     BOOKING OF INMATE
      *
      ***************************************************************
32          QUEUE       BOOKI           WAIT FOR BOOKING OFFICER
33          SEIZE       BOOK2           PROVIDE DATA TO BOOKING OFFICER
34          DEPART      BOOKI
35          ADVANCE     5,2             BOOKING PROCESS
36          RELEASE     BOOK2           BOOKING FINISHED
      ***************************************************************
      *
      *     ID PICTURES AND FINGERPRINTING
      *
      ***************************************************************
37          QUEUE       ID              WAIT FOR IDENTIFICATION PROCESS
38          SEIZE       IDOFF           GET ID OFFICER
39          DEPART      ID
40          ADVANCE     7,3            PICTURE-TAKING AND FINGERPRINTING
41          RELEASE     IDOFF           IDENTIFICATION PROCESS FINISHED
      ***************************************************************
      *
      *     PHONE CALLS
      *
      ***************************************************************
42          QUEUE       PLINE           WAIT FOR PHONES
43          ENTER       PHONE           GET A PHONE
44          DEPART      PLINE
45          ADVANCE     8,FN1           MAKE PHONE CALLS
46          LEAVE       PHONE           PHONE-CALLS FINISHED
      ***************************************************************
      *
      *     PRE-TRIAL RELEASE AND T,A,S,C, INTERVIEW
      *
      ***************************************************************
47          QUEUE       PTR             WAIT FOR PRE-TRIAL RELEASE INTERVIEW
48          ENTER       PTRO            GET AN INTERVIEWER
49          DEPART      PTR
50          ADVANCE     20,5            INTERVIEW
51          LEAVE       PTRO            INTERVIEW FINISHED
52          QUEUE       RHOLC           BACK TO REAR HOLDING CELLS
53          PRIORITY    0               LOWER PRIORITY FOR NEXT STEP
```

```
54          ENTER       REARO           GET AN OFFICER FOR INMATE MOVEMENT
55          DEPART      RHOLC
56          ADVANCE     1               MOVEMENT TIME
    ************************************************************************
    *
    *      PROPERTY ROOM
    *
    ************************************************************************
57          QUEUE       PROP            CHECK IN INMATE PROPERTY
58          SEIZE       PROPO           GET PROPERTY-ROOM OFFICER
59          DEPART      PROP
60          ADVANCE     5,2
61          RELEASE     PROPO           FINISHED WITH PROPERTY-ROOM OFFICER
    ************************************************************************
    *
    *      HOLDING CELLS 1A AND 1B
    *
    ************************************************************************
62          ADVANCE     1               MOVE INMATE TO CELLS 1A OR 1B
63          LEAVE       REARO           OFFICER AVAILABLE FOR OTHER DUTIES
64          TABULATE    1               COLLECT TRANSIT TIME STATISTICS
65          TERMINATE
    ************************************************************************
    *
    *   RUN SIMULATION FOR ONE SHIFT THREE TIMES
    *
    ************************************************************************
66          GENERATE    480             END OF SHIFT
67          TERMINATE   1
            INITIAL                     START THE SIMULATION
            START       3,,1            RUN SIMULATION FOR THREE SHIFTS
            END
```